
1 VB6 Versionen und Funktionsumfang

Es gibt aktuell drei VB6-Versionen und eine VB5 Version:

1.1.1 VB6 Learning - Edition

Von MS bewußt heruntergestuft zur Billig-Version, damit sich die teureren Versionen besser verkaufen.

Enthält den vollen VB6 Sprachumfang, sowie eine umfangreiche Palette mit Standardsteuerelementen wie

· Webbrowser

· Datentabelle

· Dateilistenfeld 

· Laufwerkslistenfeld

· Standarddialogfeld

· Kontrollkästchen

· Kombinationsfeld 

· u.v.m.

Kann eigenständig lauffähige EXE-Dateien erzeugen (Naja, braucht natürlich die VB-Runtime DLL, was die C- und Delphi-Programmierer gerne als Nachteil ins Feld führen, aber zeigt mir mal den C-Programmierer der ohne ein Betriebssystem auskommt – da nehmen sich doch die 1,3 MB direkt lächerlich aus.)

1.1.2 VB6 Professional - Edition

Wie Learning-Edition jedoch zusätzlich:

· Systemcode-Compiler der die Compiler-Technologie von Microsoft Visual C++ nutzt. 

· Unterstützung von Anwendungen, die auf Internet Information Server basieren. 

· DHTML-Seiten-Designer 

· OLE DB-Provider für Microsoft SQL Server 6.5+ und Oracle 7.3.3+ Datenbanken. 

· ActiveX Data Objects (ADO) sowie zahlreiche Steuerelemente. 

· Datenumgebungs-Designer. 

· Erstellung eigener ActiveX-DLL’s und OCX’e

· Erstellung benutzerdefinierter Datenquellen 

· Erstellung von Berichten 

· Visual Component Manager 

· Microsoft Transaction Server 2.0 

· Microsoft Internet Information Server 4.0, Developer Edition 

· Visual Database Tools Professional Edition

· Microsoft Transaction Server Debugging 

1.1.3 VB6 Enterprise - Edition

Enthält alles der niedrigeren Versionen, sowie zusätzlich:

· Visual Database Tools Enterprise Edition (Bessere SQL-Server und Oracle Unterstützung)

· Datenobjekt-Assistent 

· Visual Modeler 

· Systemeigene DEC Alpha-Unterstützung 

· Transact SQL (TSQL) Debugger 

· Microsoft SQL Server 6.5, Developer Edition 

· Microsoft Visual SourceSafe 6.0 

VB 5 CCE

Die Control Creation Edition erlaubt das erstellen von ActiveX - Steuerelementen, - DLL’s und - EXE-Dateien – allerdings keine eigenständig lauffähigen EXE-Dateien. Eine Aktualisierung ist von Microsoft meines Wissens nicht geplant, der Funktionsumfang entspricht der Learning-Edition von VB 5.

Ach ja – und im übrigen ist Sie umsonst, und auf der beiliegenden CD enthalten.

2 Vergleich der Entwicklungsumgebungen

Die IDE von VB6 entspricht im wesentlichen der Vorgängerversion von VB 5 bzw. den VBA-Applikationen Word 97 und Excel 97. Access hat nun mit der 2000 auch gleichgezogen, ob dies nun einen Vor- oder Nachteil darstellt, ist natürlich Geschmackssache, ich persönlich bevorzuge die Entwicklungsumgebung von Access 97, da die dockable Windows die Arbeitsfläche erheblich einschränken (läßt sich aber auch abschalten). 

Die Hilfeunterstützung durch das neue HTML-Help stellt m.E. auch eine Verschlimmbesserung dar, eine gedruckte Dokumentation sucht man selbst im teueren Enterprise-Paket vergebens (dafür aber eine Broschüre mit der Aufschrift „Woran Sie Original-Software von Microsoft erkennen – an Originalhandbüchern“), diese kann man aber für schlappe DM 300. - nachbestellen.

Wesentliche Unterschiede ergeben sich durch die unterschiedlichen Eigenschaften der Formulare, Steuerelemente und Klassenmodule:


VB 6
Access 97
Access 2000

Formulare
Sind generell nicht datengebunden; die Datenbankanbindung erfolgt in jedem einzelnen Steuerelement durch Einstellung der DataMember-Eigenschaft

+ Freie Wahl zwischen MDI- und SDI
Alle Formulare können datengebunden sein

- Generell MDI
Wie in 97

Berichte
Die Datenbankanbindung erfolgt wie bei den Formularen
Alle Berichte können datenbankgebunden sein
Wie in 97, erweiterte Zeilenformatierung

Standardsteuerelemente
Generell Datenbankanbindung möglich

+ Erweiterte Einstellungsmöglichkeiten über API-Funktionen
Generell Datenbankanbindung möglich
Generell Datenbankanbindung möglich

Tabellen
MS-Datagrid

MS-Flexgrid

+ Umfangreiche Programmierbarkeit
Standardformular

+ Sehr umfangreich und komfortabel

- Eingeschränkte Programmierbarkeit 
Wie in 97, erweiterte Zeilenformatierung

+ Äquivalent zum Flexgrid

List- und Comboboxen
MS-Datalist u. Datacombo

- eingeschränkter Funktionsumfang gegenüber den MS-Access Äquivalenten
List- und Comboboxen

+ Sehr umfangreich und komfortabel
Wie in 97

+ Sehr umfangreich und komfortabel

Zusatzsteuerelemente
Treeview, Listview, Drivelistbox, Filelistbox, Dirview, MSComm-Control etc.
Nur Developer-Edition
Nur Developer-Edition

Menüs
Nur Formulargebunden

+ Leichtere Programmierbarkeit

- Keine Icons in Menüs
Nur IDE-Gebunden

+ Icons und Comboboxen in Menüs möglich

- Umständliche Programmierung
Nur IDE-Gebunden

+ Icons und Comboboxen in Menüs möglich

- Umständliche Programmierung

Register
MS-Tab

-Unkomfortabel, da kein Container

Sheridan Tabbed Control

+ Erweiterte Layout-Möglichkeiten; sehr komfortabel
Register-Steuerelement

+ Komfortabel
Register-Steuerelement

+ Komfortabel

2.1 Einsatzmöglichkeiten von VB

Die VB-Entwicklungsumgebung positioniert sich nach MS als RAD-Tool (Rapid Application Development) und ist m.E. das Schweizer-Offiziersmesser unter den Entwicklungsumgebungen – Man kann fast alles damit machen, aber nichts besonders gut. Dies hört sich im ersten Moment natürlich etwas härter an als es ist, deswegen möchte ich diese Aussage im folgenden etwas relativieren, und die Möglichkeiten und Einschränkungen von VB etwas erläutern:

2.1.1 VB zur Entwicklung von Branchenlösungen

Der oft genannte Vorteil von VB - Kann EXE’s erzeugen – beinhaltet auch die Schwäche für diesen Bereich: Mit VB entwickelte Anwendungen erfordern eine weitaus größere Entwicklungszeit und Aufwand als die mit Access entwickelten. 

Kann man bei Access geschützte Anwendungen vertreiben, die dem Endbenutzer ein hohes Maß an Eingriffen ermöglichen wie z.B. die Formatierung von Tabellenansichten, den Entwurf eigener Abfragen etc. scheidet das bei VB völlig aus, der Entwickler muß alle diese Möglichkeiten bereits im Entwurf der Anwendung berücksichtigen. Auch bei der Behandlung von Laufzeitfehlern verzeiht die MS-Access-Runtime doch manchen Fehler, was bei VB unbarmherzig zur Terminierung der Anwendung führt.

Hat man diese Hürden gemeistert, kann man mit VB natürlich Anwendungen erstellen, die am Funktionsumfang und auch in der Oberflächengestaltung wesentlich mehr Möglichkeiten bieten als vergleichbare Access-Anwendungen, auch kann man sich durch den Einsatz von OCX’e und Tools von Drittherstellern, die mit VB vergleichsweise (im Vergleich zu Access) problemlos laufen, einiges an Entwicklungszeit ersparen.

Auch die Möglichkeiten von VB zur Einbindung eigener Usercontrols (dazu später mehr) und Funktionsbibliotheken erlauben eine leichte Wartbarkeit und hohes Maß an wiederverwendbarem Code.

VB 6 verglichen mit Access 97 erlaubt eine bessere Anbindung an SQL-Server, die verwendete Jet-Engine entspricht mit der Version 4.0 der von Access 2000, welche nun auch eine hervorragende SQL-Server-Unterstützung bietet. Angesichts jedoch des Installationsumfangs, kann man mit VB natürlich wesentlich „schlankere“ Lösungen anbieten – was soll das, werden jetzt vielleicht einige einwenden, ich liefer doch eh auf CD aus, aber im Zeitalter des Internets ist es auch im Hinblick auf Online-Vertrieb doch ein erheblicher Unterschied, ob der Kunde sich die Demo einfach runterlädt oder erst bestellen muß – was unter Umständen zu einem erheblichen Wettbewerbsvorteil führt.

Zusammenfassend würde ich VB als das Werkzeug zur Entwicklung von Branchenlösungen mit einer breiten Installierbaren Basis empfehlen, MS-Access bleibt für mich als das Mittel zur leichten Erstellung von grundlegenden Konzeptentwürfen und zur Entwicklung von Individuallösungen.

2.1.2 VB zur Entwicklung von ActiveX – Steuerelementen und DLL’s

Bereits mit der kostenlos verfügbaren VB5 CCE kann man ActiveX-Steuerelemente und DLL’s erzeugen, VB6 bietet darüber hinaus die Möglichkeit echte datengebundene Steuerelemente zu erzeugen. 

Dies erlaubt die leichtere Wiederverwendbarkeit von Code in den eigenen Anwendungen, da man die Versionskontrolle auch selber ausübt, hat man die bekannten Probleme mit den MS-OCX’e nicht, die zu einer allgemeinen Verunsicherung in der Access-Gemeinde zum Einsatz von ActiveX-Steuerelementen geführt haben.

2.1.3 VB zur Entwicklung von COM-AddIn’s

Mit Access 2000 hat MS eine einheitliche Entwicklungsumgebung für alle Office-Produkte eingeführt, was die Programmierbarkeit von Access zwar m.E. nur verschlimmbessert hat, aber die Möglichkeit erlaubt, mit VB ein Add-In für   a l l e  Office – IDE’s  verfügbar zu machen – und Add-Ins für die Einzelapplikationen zu entwickeln.

2.2 Konvertierungs-Tools

2.2.1 Access To VB 4.11

AccessToVB ist realisiert als VB-Add-In, in der Version 4.11 konvertiert es nur MS-Access 97 Datenbanken, es steht aber bereits eine neuere Version zur Verfügung, die auch Access 2000 Datenbanken konvertiert.

Die Demoversion konvertiert ausschließlich die Beispiel-Datenbank Nordwind, wie ein eigener Test ergab, kann sich das Ergebnis trotz einer notwendigen manuellen Nachbearbeitung durchaus sehen lassen. Selbst komplexe Unterformulare werden in Usercontrols zur Verwendung in dem MS-Datarepeater-Steuerelement umgewandelt.

Wenn man gewisse Vorarbeiten leistet und auch Rücksicht auf die Namenskonventionen nimmt, was MS in seinen eigenen Beispiel-Datenbanken ja leider nicht für nötig hält, kann man sich erheblichen Nachbearbeitungsaufwand sparen.

2.2.2 Access Upsizer

Der Access Upsizer ist in der mir vorliegenden Version etwas veraltet, da er auf VB 5 abgestimmt ist und damit nur den Datenzugriff über DAO unterstützt. Ein Test war leider nicht möglich, da er wie AccessToVB in der Demoausführung nur die Nordwind-Datenbank konvertiert und offenbar nur mit der englischen Version dieser Beispieldatei zurechtkommt, welche mir nicht zur Verfügung stand.

2.2.3 Object Converter

Der Object-Converter ist eine Access 97 – Add-In Datenbank, die ausgewählte Datenbankobjekte in die entsprechende VB-Datei konvertiert. Die Demoversion ist beschränkt auf 6 Konvertierungen, meine Versuche mit diesem Tool zeigten, das dieses etwas Probleme mit Registersteuerelementen hat, aber für kleinere Konvertierungsarbeiten mit etwas Nachbearbeitungsaufwand durchaus brauchbar ist.

2.3 ADO im Vergleich mit DAO

Die von MS als Nachfolger von DAO und ODBC propagierte Datenbankschnittstelle ist unter dem Oberbegriff „OLE-DB“ zusammengefaßt, wobei es sich um weitaus mehr handelt als nur um eine „weitere oder neue“ Datenbankschnittstelle. Sie stellt vielmehr den Universellen Datenzugriff für a l l e Formen von „Daten“ bereit, egal ob nun in einer beliebigen Datenbank gespeichert, oder einem Filesystem oder ähnlichem.

Sie gliedert sich funktional in ein 3 Schichtenmodell, die aus der grundlegender Datenzugriffschicht besteht, die die Verbindung zur Anwendung darstellt, und aus den Mittlern zwischen Anwendung und Services, die durch verschiedene Serviceprovider den Zugriff zu den einzelnen Datenbanken darstellt.

Da ADO eine weitaus universellere Datenschnittstelle bietet als das altbekannte DAO, soll im folgenden nur auf den Datenbankzugriff mittels des Jet-Providers auf MS-Access Datenbanken eingegangen werden.

Die Datenbankschnittstelle, die zuvor im DAO – Objektmodell vereinigt war, gliedert sich nun in 3 Objektmodelle:

2.3.1 ADO

Bibliothek: „Microsoft ActiveX Data Objects 2.1 Library."

ADO erlaubt Ihrer Anwendung mittels verschiedener OLE-DB Provider auf beliebige Datenquellen zuzugreifen, um Datensätze zu lesen, hinzuzufügen, bearbeiten und zu löschen. 

2.3.2 ADOX

Bibliothek: Microsoft ADO Ext. 2.1 for DDL and Security

Das ADOX-Objektmodell enthält Objekte für Datenbankobjektmanipulation (Tabellen, Abfragen und Indexe) und Erstellung von Benutzern und Gruppen sowie die Zuweisung von Rechten an Datenbankobjekten von Benutzern und Gruppen.

Sowohl ADO als auch ADOX benutzen das Connection-Objekt als Grundlage eines atenbankzugriffes, das Catalog-Objekt ist der Container für die Datenbankobjekte (Tabellen, Abfragen etc.) und die Sicherheitsobjekte (Benutzer und Gruppen). Jedes Catalog-Objekt ist verknüpft mit einem einzigen Connection-Objekt als Grundlage der Datenbankanbindung.

Jedes Tabellen-, Index, und Spalten-Objekt hat ebenso eine Standard ADO-Properties Collection.

2.3.3 JRO

Bibliothek: Microsoft Jet and Replication Objects 2.1 Library

Das JRO Objektmodell enthält Funktionen, Objekte und Eigenschaften zum Erstellen, Bearbeiten und Synchronisieren von Replikaten und kann ausschließlich mit Jet-Datenbanken verwendet werden.

Es enthält ferner Funktionen zum Komprimieren von Jet-Datenbanken.

2.3.4 Codebeispiele

Im folgenden werden DAO und ADO bzw. ADOX Methoden gegenübergestellt.

2.3.4.1 Eine Datenbank öffnen

DAO
Sub DAOOpenJetDatabase()

    Dim db      As DAO.Database

    

    Set db = DBEngine.OpenDatabase("C:\Nwind.mdb")

    db.Close

    

End Sub

ADO
Sub ADOOpenJetDatabase()

    Dim cnn    As New ADODB.Connection

    

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\Nwind.mdb;"

    cnn.Close

    

End Sub

2.3.4.2 Datenbank-Properties einstellen

DAO
Sub DAOSetJetDBOption()

    Dim db      As DAO.Database

    

    Set db = DBEngine.OpenDatabase("C:\nwind.mdb")

    DBEngine.SetOption dbPageTimeout, 4000

    db.Close

    

End Sub

ADO
Sub ADOSetJetDBOption()

    Dim cnn    As New ADODB.Connection

    

    cnn.Provider = "Microsoft.Jet.OLEDB.4.0"

    cnn.Open "C:\nwind.mdb"

    cnn.Properties("Jet OLEDB:Page Timeout") = 4000

    cnn.Close

End Sub

2.3.4.3 Passwortgeschützte Datenbank öffnen

DAO
Sub DAOOpenDBPasswordDatabase()

    Dim db      As DAO.Database

    

    Set db = DBEngine.OpenDatabase("C:\nwind.mdb", False, False, _

        ";pwd=password")

    db.Close

End Sub

ADO
Sub ADOOpenDBPasswordDatabase()

    Dim cnn      As New ADODB.Connection

    

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=C:\nwind.mdb;Jet OLEDB:Database Password=password;"

    cnn.Close

    

End Sub

2.3.4.4 Einen Verweis auf CurrentDB setzen

Das folgende Beispiel funktioniert nur unter Access 2000, da in VB6 der Designer mehrere Connections gleichzeitig öffnen kann.

DAO
Sub DAOGetCurrentDatabase()

    Dim db      As DAO.Database

    

    Set db = CurrentDb()

    

End Sub

ADO
Sub ADOGetCurrentDatabase()

    Dim cnn     As ADODB.Connection

    

    Set cnn = CurrentProject.Connection

    

End Sub

2.3.4.5 Datensatzgruppen bearbeiten

Sowohl DAO als auch DAO verwenden das Recordset-Objekt, diese sind aber nicht kompatibel. Die Zuweisung eines DAO.Recordset zu einem ADO.Recordset muß daher unweigerlich fehlschlagen.

DAO
Sub DAOOpenRecordset()

    Dim db      As DAO.Database

    Dim rst     As DAO.Recordset

    Dim fld     As DAO.Field

    

    'Open the database

    Set db = DBEngine.OpenDatabase("C:\nwind.mdb")

    

    'Open the Recordset

    Set rst = db.OpenRecordset("Select * from Customers where Region" & _

        " = 'WA'", dbOpenForwardOnly, dbReadOnly)

    

    ' Print the values for the fields in the first record in the debug

    ' window

    For Each fld In rst.Fields

        Debug.Print fld.Value & ";";

    Next

    'Close the recordset

    rst.Close

End Sub

ADO
Sub ADOOpenRecordset()

    Dim cnn     As New ADODB.Connection

    Dim rst     As New ADODB.Recordset

    Dim fld     As ADODB.Field

    

    ' Open the connection

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\nwind.mdb;"

    

    ' Open the forward-only, read-only recordset

    rst.Open "Select * from Customers where Region = 'WA'", cnn, _

        adOpenForwardOnly, adLockReadOnly

    

    ' Print the values for the fields in the first record in the debug

    ' window

    For Each fld In rst.Fields

        Debug.Print fld.Value & ";";

    Next

    'Close the recordset

    rst.Close

    

End Sub

2.3.4.6 Einen Datensatz finden

DAO enthält vier Methoden: FindFirst, FindLast, FindNext, FindPrevious. Sie müssen zu Beginn auswählen, in welcher Richtung Sie suchen wollen, und von wo aus sie suchen.

ADO hat eine eizige Methode: Find. Die Suche beginnt immer vom aktuellen Datensatz. Die Find Methode hat Parameter die es erlaubt die Suchrichtung zu bestimmen.

In DAO gibt es die Eigenschaft NoMatch des Recordset Objektes, die bestimmt ob die Suche erfolgreich war oder nicht. In ADO gibt es diese nicht, wenn die Suche fehlschlägt ist die Position des Datensatzzeiger entweder EOF (Suchrichtung vorwärts - adSearchForward ) oder BOF (Suchrichtung rückwärts - adSearchBackward).

DAO und ADO verwenden eine unterschiedliche Syntax um Datensätze zu finden, die einen Nullwert im Feld enthalten. In DAO verwenden Sie folgende Syntax um einen Datensatz mit einem Nullwert in einem Feld zu finden:

"ColumnName Is Null"

oder auch 

"ColumnName Is Not Null"

ADO, aus welchen Gründen auch immer, kennt keinen Is-Operator. Sie müssen daher den = oder <> Operator verwenden:
"ColumnName = Null"

oder auch 

"ColumnName <> Null"

DAO 
Sub DAOFindRecord()

    Dim db      As DAO.Database

    Dim rst     As DAO.Recordset

            

    ' Open the database

    Set db = DBEngine.OpenDatabase("C:\nwind.mdb")

    

    ' Open the Recordset

    Set rst = db.OpenRecordset("Customers", dbOpenDynaset)

    

    ' Find the first customer who's country is USA

    rst.FindFirst "Country = 'USA'"

    

    ' Print the customer id's of all customers in the USA

    While Not rst.NoMatch

        Debug.Print rst.Fields("CustomerId").Value

        rst.FindNext "Country = 'USA'"

    Wend

    

    ' Close the recordset

    rst.Close

    

End Sub

ADO
Sub ADOFindRecord()

    Dim cnn     As New ADODB.Connection

    Dim rst     As New ADODB.Recordset

    

    ' Open the connection

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\nwind.mdb;"

    

    ' Open the recordset

    rst.Open "Customers", cnn, adOpenKeyset, adLockOptimistic

    

    ' Find the first customer who's country is USA

    rst.Find "Country='USA'"

    

    ' Print the customer id's of all customers in the USA

    While Not rst.EOF

        Debug.Print rst.Fields("CustomerId").Value

        rst.Find "Country='USA'", 1

    Wend

    

    ' Close the recordset

    rst.Close

    

End Sub 

2.3.4.7 Eine Datenbank erstellen

DAO
Sub DAOCreateDatabase()

    Dim db      As DAO.Database

    Set db = DBEngine.CreateDatabase("C:\new.mdb", dbLangGeneral)

    

End Sub

ADOX
Sub ADOCreateDatabase()

    Dim cat     As New ADOX.Catalog

    

    cat.Create "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=C:\new.mdb;"

End Sub

2.3.4.8 Auf Datenbankobjekte zugreifen

DAO
Sub DAOListTables()

    Dim db      As DAO.Database

    Dim tbl     As DAO.TableDef

    

    ' Open the database

    Set db = DBEngine.OpenDatabase("C:\nwind.mdb")

    ' Loop through the tables in the database and print their name

    For Each tbl In db.TableDefs

        Debug.Print tbl.Name

    Next

End Sub

ADOX
Sub ADOListTables()

    Dim cat     As New ADOX.Catalog

    Dim tbl     As ADOX.Table

    ' Open the catalog

    cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=c:\nwind.mdb;"

    ' Loop through the tables in the database and print their name

    For Each tbl In cat.Tables

        If tbl.Type <> "VIEW" Then Debug.Print tbl.Name

    Next

    

End Sub

2.3.4.9 Das Passwort eines Benutzers ändern

DAO
Sub DAOChangePassword()

    Dim wks     As Workspace

    Dim usr     As DAO.User

    

    ' Open the workspace, specifying the system database to use

    

    DBEngine.SystemDB = "C:\sysdb.mdw"

    Set wks = DBEngine.CreateWorkspace("", "Admin", "")

    

    ' Change the password for the user Admin

    wks.Users("Admin").NewPassword "", "password"

End Sub

ADOX
Sub ADOChangePassword()

    Dim cat     As New ADOX.Catalog

    

    ' Open the catalog, specifying the system database to use

    cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=C:\nwind.mdb;Jet OLEDB:System database=C:\sysdb.mdw"

    

    ' Change the password for the user Admin

    cat.Users("Admin").ChangePassword "", "password"

    

End Sub

2.3.4.10 Mit dem Error-Objekt arbeiten

DAO
Sub DAODatabaseError()

On Error GoTo DAODatabaseError_Err

    Dim db      As DAO.Database

    Dim errDB   As DAO.Error

    

    Set db = DBEngine.OpenDatabase("c:\nonexistant.mdb")

    

    Exit Sub

    

DAODatabaseError_Err:

    For Each errDB In DBEngine.Errors

        Debug.Print "Description: " & errDB.Description

        Debug.Print "Number: " & errDB.Number

        Debug.Print "JetErr: " & errDB.Number

    Next

End Sub

ADO
Sub ADODatabaseError()

On Error GoTo ADODatabaseError_Err

    Dim cnn     As New ADODB.Connection

    Dim errDB   As ADODB.Error

    

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;" & _

      "Data Source=c:\nonexistant.mdb"

    

    Exit Sub

    

ADODatabaseError_Err:

    For Each errDB In cnn.Errors

        Debug.Print "Description: " & errDB.Description

        Debug.Print "Number: " & errDB.Number

        Debug.Print "JetErr: " & errDB.SQLState

    Next

End Sub

2.3.4.11 Eine Datenbank komprimieren

DAO
Sub DAOCompactDatabase()

    ' Make sure there isn't already a file with the

    ' name of the compacted database.

    If Dir("c:\newnwind.mdb") <> "" Then _

        Kill "c:\newnwind.mdb"

    

    ' Basic compact - creating new database named newnwind

    DBEngine.CompactDatabase "C:\nwind.mdb", "C:\newnwind.mdb"

    

    ' Delete the original database

    Kill "c:\nwind.mdb"

    

    ' Rename the file back to the original name

    Name "c:\newnwind.mdb" As "c:\nwind.mdb"

    

End Sub

JRO
Sub JROCompactDatabase()

    Dim je      As New JRO.JetEngine

        

    ' Make sure there isn't already a file with the

    ' name of the compacted database.

    If Dir("c:\newnwind.mdb") <> "" Then _

        Kill "c:\newnwind.mdb"

    

    ' Compact the database

    je.CompactDatabase "Data Source=C:\nwind.mdb;", _

        "Data Source=C:\newnwind.mdb;"

        

    ' Delete the original database

    Kill "c:\nwind.mdb"

    

    ' Rename the file back to the original name

    Name "c:\newnwind.mdb" As "c:\nwind.mdb"

    

End Sub

2.3.5 Neue Funktioen in ADO, ADOX, and JRO
Der folgende Abschnitt beschreibt einige der neuen Funktionen von ADO, ADOX, JRO und dem Jet Provider. 

2.3.5.1 Recordset-Objekte anstatt Arrays oder Collections verwenden

Anstatt eines Arrays oder eines Collection-Objektes können Sie ein Recordset-Objekt per Code erzeugen, mit Daten füllen, bearbeiten und speichern.

Sub ADOCreateRecordset()

    Dim rst As New ADODB.Recordset

    rst.CursorLocation = adUseClient

    'Add Some Fields

    rst.Fields.Append "dbkey", adInteger

    rst.Fields.Append "field1", adVarChar, 40, adFldIsNullable

    rst.Fields.Append "field2", adDate

    

    'Create the Recordset

    rst.Open , , adOpenStatic, adLockBatchOptimistic

    

    'Add Some Rows

    rst.AddNew Array("dbkey", "field1", "field2"), _

       Array(1, "string1", Date)

    rst.AddNew Array("dbkey", "field1", "field2"), _

       Array(2, "string2", #1/6/1992#)

    

    'Look at the values - a value of 1 for status column = newly record

    rst.MoveFirst

    Debug.Print "Status", "dbkey", "field1", "field2"

    While rst.EOF <> True

       Debug.Print rst.Status, rst!dbkey, rst!field1, rst!field2

       rst.MoveNext

    Wend

    

    'Commit the rows without ActiveConnection set resets the status bits

    rst.UpdateBatch adAffectAll

    

    'Change the first of the two rows

    rst.MoveFirst

    rst!field1 = "changed"

    

    'Now look at the status, first row shows 2 (modified row),

    'second shows 8 (no modifications)

    'Also note that the OriginalValue property shows the value

    'before the modification

    rst.MoveFirst

    While rst.EOF <> True

       Debug.Print

       Debug.Print rst.Status, rst!dbkey, rst!field1, rst!field2

       Debug.Print , rst!dbkey.OriginalValue, _

          rst!field1.OriginalValue, rst!field2.OriginalValue

       rst.MoveNext

    Wend

End Sub

2.3.5.2 Auflistung der angemeldeten Benutzer erhalten

Sub UserRoster()

' List all of the users that are currently logged into the database

    

    Dim cnn     As New ADODB.Connection

    Dim rst     As ADODB.Recordset

    

    ' Open the connection

    cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\nwind.mdb;"

    

    ' Open the user roster schema rowset

    Set rst = cnn.OpenSchema(adSchemaProviderSpecific, , _

        JET_SCHEMA_USERROSTER)

    

    ' Print the results to the debug window

    Debug.Print rst.GetString()

    

    cnn.Close

    

End Sub

2.3.5.3 Autowert-Felder bearbeiten

Sub ADOCreateEnhancedAutoIncrColumn()

    Dim cat     As New ADOX.Catalog

    Dim col     As New ADOX.Column

    

    ' Open the catalog

    cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

            "Data Source=C:\nwind.mdb;"

        

    ' Create the new auto increment column

    With col

        .Name = "ContactId"

        .Type = adInteger

        Set .ParentCatalog = cat

        .Properties("AutoIncrement") = True

        .Properties("Seed") = CLng(10)

        .Properties("Increment") = CLng(100)

    End With

    

    ' Append the column to the table

    cat.Tables("Contacts").Columns.Append col

    

    Set cat = Nothing

End Sub

� Microsoft Visual Basic Produktinfo: � HYPERLINK "http://www.eu.microsoft.com/germany/msdn/vbasic/default.htm" ��http://www.eu.microsoft.com/germany/msdn/vbasic/default.htm�


� Access To VB - Deutscher Vertrieb unter � HYPERLINK "http://www.accesstovb.de/" ��http://www.accesstovb.de/�


� Migrating from DAO to ADO: � HYPERLINK "http://msdn.microsoft.com/library/sdkdoc/dasdk/migr4yun.htm" ��http://msdn.microsoft.com/library/sdkdoc/dasdk/migr4yun.htm�

